

CERN – European Organization for Nuclear Research
European Laboratory for Particle Physics



CTF3 Note 023 (Tech.)
(Delay Loop, Beam Optics)

DRIVING BEAM DELAY LOOP DESIGN FOR CTF3

F. Sannibale

Abstract

In CLIC (Compact Linear Collider) the 30 GHz RF power, necessary for accelerating the electron and positron beams in the main linac structures, must be produced by the *driving beam*. This electron beam with high charge per bunch and with a time structure of 15 GHz passes through special decelerating structures that extract the 30 GHz component to be used for the main beam acceleration. In the CLIC Test Facility CTF2 the basic principles of this RF power production have been demonstrated. In CTF3 [1] the feasibility of the compression scheme that allows passing from a bunch separation of 20 cm to the final one of 2 cm of the 15 GHz time structure will be checked. The necessary compression is obtained in two steps: a factor two in the Delay Loop (DL) and a factor five in the Combiner Ring [2]. In this note the design of the DelayLoop is presented.

Geneva, Switzerland
18 June 2001

CTFF3 TECHNICAL NOTE

INFN - LNF, Accelerator Division

Frascati, April 4, 2001

Note: **CTFF3-001**

DRIVING BEAM DELAY LOOP DESIGN FOR CTF-3

F. Sannibale

In CLIC (Compact Linear Collider) the 30 GHz RF power, necessary for accelerating the electron and positron beams in the main linac structures, must be produced by the *driving beam*. This electron beam with high charge per bunch and with a time structure of 15 GHz passes through special decelerating structures that extract the 30 GHz component to be used for the main beam acceleration. In the CLIC Test Facility CTF2 the basic principles of this RF power production have been demonstrated. In CTF3 [1] the feasibility of the compression scheme that allows passing from a bunch separation of 20 cm to the final one of 2 cm of the 15 GHz time structure will be checked. The necessary compression is obtained in two steps: a factor two in the Delay Loop (DL) and a factor five in the Combiner Ring [2]. In this note the design of the Delay Loop is presented.

1. INTRODUCTION

CTF3 will be installed at CERN in the existing buildings of the LEP pre-injector accelerators LIL and EPA. A relevant part of the components such as klystrons, magnets and power supplies of LIL and EPA will be reused when possible.

The CTF3 complex starts with a 3 GHz linac that produces a pulsed electron beam with macrobunches 1.4 μ sec long and with energy tunable from 180 MeV up to 350 MeV. The separation between the bunches at the linac output is 20 cm (twice the linac RF wavelength) so that the macrobunch time structure alternatively presents empty and filled buckets. Moreover the macrobunch is composed by alternated sequences 140 ns long of *even* and *odd* buckets, the difference in phase between them being one RF wavelength. The linac is connected by a transfer line to the DL where a 1.5 GHz RF deflector, see Figure 1, deviates on the left the odd bunches sequences inside the DL and the even ones on the right. The DL length is 140 ns times c so that after the DL the odd sequence will be recombined with the incoming even sequence in such a way to fill the interleaved empty buckets. The resulting macrobunch structure, at the DL output, presents 140 ns long trains of buckets separated now by 10 cm, followed by 140 ns long voids.




Figure 1. Delay Loop Injection Scheme.

The general characteristics of the DL and the described recombination scheme with the single RF deflector were first presented in [1]. Table 1 shows a summary of the main requirements and constraints that the DL must fulfill.

Table 1. Delay Loop Main Requirements and Constraints.

Length [m]	42
Energy [MeV]	160 ÷ 350
Maximum Charge/Bunch [nC]	2.3
Isochronicity	$ \alpha_p \leq 5 \cdot 10^{-4}$
Momentum Acceptance (total)	$\Delta p/p = 5\%$
Path Length Tuning [mm]	± 0.5

2. DELAY LOOP DESCRIPTION

Figure 2 shows the DL layout while tables 1, 2 and 3 report the DL main parameters. The MAD8 Input file of the DL can be found in the Appendix to the note. According to the CTF3 philosophy of reusing when possible existing hardware, design material and documentation, the great part of the DL magnetic components, see Table 2, are copies of already existing magnets. In fact, the bending magnets are the 22;5 rectangular dipoles (HR.BHZij) of EPA [3] now used as 30; magnets, while the 2; DC septa magnets, the quadrupoles and the sextupoles are copies of magnets of the DAΦNE Accumulator Ring [4, 5, 6]. The 27;14 DC septa are a modified version of the 34; DAΦNE Accumulator septa. New elements are the single period wiggler used for tuning the DL length (same as the ones used in the CTF3 Combiner Ring [2]) and the very small 5 mrad dipoles used in the injection region.




Figure 2. Delay Loop Layout.

Table 2. Delay Loop Magnetic Elements.

A) Number of Dipoles (EPA-like)	10
Dipole Bending Radius [m]	1.079
Dipole Bending Angle [deg]	30
Dipole Field [T] (180/350 MeV)	0.60/1.08
Integrated Quadrupole Coeff. in Dipoles [T] (180/350 MeV)	0.22/0.43
B) Number of Quadrupoles (DAΦNE Accumulator-like)	34
Max. Integrated Gradient [T] (180/350 MeV)	1.42/2.73
Quadrupole Families (minimum/preferred)	14/18
C) Number of Sextupoles (DAΦNE Accumulator-like)	10
Max Integrated Gradient [T/m] (180/350 MeV)	11.4/22.2
Sextupole Families	5
D) Number of Path Length Tuning Wiggler	1
E) Number of 2 _i Septa (DAΦNE Accumulator-like)	2
F) Number of 27 _i .14 Septa (DAΦNE Accumulator-like)	2
G) Injection Dipoles Number	2
Injection Dipoles Bending Angle [mrad]	5

2.1. Lattice

In designing the DL an important constraint is the isochronicity of the lattice. This condition must be carefully fulfilled in order to preserve the longitudinal structure of the bunch along the loop. To the first order, the particle longitudinal position in the bunch is given by:

$$ct = (ct)_0 + R_{56} \frac{\Delta p}{p} \quad (1)$$

where c is the speed of light, $\Delta p/p$ the momentum deviation and R_{56} is the momentum related term of the DL first order transfer matrix. Isochronicity is obtained when such a term vanishes:

$$R_{56} = \int \frac{D_x}{\rho} ds = 0 \quad (2)$$

According to expression (2) the horizontal dispersion D_x must be properly arranged inside the DL dipoles where the trajectory curvature radius ρ assumes a finite value. At the same time the very large momentum acceptance requires small dispersion values all over the loop for containing beam envelopes.




Figure 3. Delay Loop Optical Functions. Half Loop.

Table 3. Delay Loop Main Parameters.

Max. Horizontal Beta [m]	14.8
Max. Vertical Beta [m]	12.7
Max. Dispersion [m]	0.85
Horizontal Betatron Tune	5.90
Vertical Betatron Tune	2.19

Figure 3 shows the optical functions, as calculated by MAD8, of half the DL being the second half exactly symmetric. The half machine can be divided in quarters each one with a triple bend arc with the sign of dispersion alternated in the dipoles. Dispersion and dispersion derivative are zero at the end and at the output of each quarter and the isochronicity condition (2) is obtained in half the machine. In the quarter including the injection the role of the first bend in the arc is performed by the couple of septa magnets. Horizontal and vertical functions are well separated in the entire loop and in particular the vertical beta assumes small values inside the dipoles and the wiggler where the reduced stay clear requires small beam sizes.

Figure 4 shows the 14 quadrupole families while in Table 4 their strength is reported. It must be reminded that the dipoles present a quadrupole component and that the model used for the bends is the one by T. Risselada (January 2001) based on the magnetic measurements reported in [7].

In order to facilitate the matching between the DL and the downstream transfer line, the number of quadrupole families could be increased to 17 or 18 by making the 3 or 4 quadrupoles before the output septa independent.




Figure 4. Quadrupole and Sextupole Families.

Table 4. Quadrupole Families Strengths.

Family Name	Quads/Famil y	L [m]	k^2 [m ⁻²]
Q1spt	2	0.3	- 2.580
Q2spt	2	0.3	+ 7.428
Q3spt	2	0.3	- 4.615
Q4spt	2	0.3	+ 4.095
Cq1	2	0.3	+ 1.302
Cq2	4	0.3	+ 7.032
Cq3	4	0.3	- 7.007
Cq4	4	0.3	+ 6.681
Cq5	2	0.3	+ 5.653
Cq6	2	0.3	- 5.140
Cq7	2	0.3	+ 7.809
Qmt1	2	0.3	+ 0.556
Qmt2	2	0.3	- 3.367
Qmt3	2	0.3	+ 4.588

The isochronicity requirement in Table 1 asks for a momentum compaction $|\alpha_p| \leq 5 \cdot 10^{-4}$. As the DL is 42 m long, this means $|R_{56}| \leq 2.1 \cdot 10^{-2}$ m. The R_{56} variation must be done maintaining the matching at the DL output and preferably without generating strong modifications in the optical functions. The best way for achieving such a situation is by performing small variations on the strength of all the DL loop quadrupole families. Table 5 shows, as example, the strength variations necessary for obtaining $R_{56} = 1.0 \cdot 10^{-2}$ m.

Table 5. Quadrupole Strength Variations for $R_{56} = 1.0 \cdot 10^{-2} m$.

Family Name	Strength Variation [%]
Q1spt	1.48
Q2spt	0.182
Q3spt	-1.52
Q4spt	0.470
Cq1	-1.59
Cq2	0.007
Cq3	-0.349
Cq4	-0.093
Cq5	-0.467
Cq6	-0.097
Cq7	0.405
Qmt1	-0.617
Qmt2	-0.105
Qmt3	-0.482

2.2. Second Order Isochronicity

Figure 5 shows a MAD8 tracking output (TRANSPORT mode). Off-energy particles within $\Delta p/p \pm 2.5 \cdot 10^{-2}$ are tracked from the DL input (crosses) up to the output (dots) and the related longitudinal positions inside the bunch are showed in the ordinate (for all the particles $x_0 = x'_0 = y_0 = y'_0 = 0$). In the DL transfer matrix isochronicity is corrected only to the first order ($R_{56} = 0 m$). The large displacement of the off-energy particles indicates that a higher order correction is necessary. To the second order:

$$\begin{aligned}
 ct = (ct)_{0+} R_{56} \frac{\Delta p}{p} + T_{516} x_0 \frac{\Delta p}{p} + T_{526} x'_0 \frac{\Delta p}{p} + T_{536} y_0 \frac{\Delta p}{p} + \\
 + T_{546} y'_0 \frac{\Delta p}{p} + T_{556} (ct)_0 \frac{\Delta p}{p} + T_{566} \left(\frac{\Delta p}{p} \right)^2
 \end{aligned} \quad (3)$$

where T_{ijk} are elements of the second order transfer matrix. Isochronicity is achieved when:

$$T_{5i6} = 0 \quad \forall i \quad (4)$$

Five families of sextupoles were added in the DL layout in order to obtain condition (4), see Figures 3 and 4.




Figure 5. Off-energy Particles Longitudinal Position at DL Input (Crosses) and Output (Dots) First Order Correction.




Figure 6. Off-energy Particles Longitudinal Position at DL Input (Crosses) and Output (Dots) First and Second Order Correction.

Figure 6 shows the tracking result with the second order correction applied, the sextupole strengths used for the second order correction are reported in Table 6. Condition (4) is fulfilled using only 4 of the 5 families. The extra family has been added in order to give some additional capability in tuning the second order matrix terms. In fact, other second order terms non-depending on the energy deviation affect the position of the particle inside the bunch and must be controlled. Moreover, the terms affecting the particle transverse position along the DL must be controlled as well for avoiding particle losses. The second order correction presented here must be considered preliminary and extensive six-dimension particle tracking is required in order to fully understand the second order behavior of the DL matrix.

Table 6. Sextupole Families Strengths.

Family Name	Sext./Family	L [m]	k [m ⁻³]
Sd0	2	0.1	- 189.7
Sd1	2	0.1	0
Sd2	2	0.1	- 69.44
Sd3	2	0.1	- 26.81
Sd4	2	0.1	- 143.4

2.3. Path Length Tuning

The wiggler magnet placed in the half machine straight section performs the DL length tuning. This compact device, already presented in reference [2], is a single period wiggler with two half end poles and a central pole independently powered. The nominal bending angles are -7_i , 14_i , -7_i and the total length is 1.6 m. The required path length tuning of ± 0.5 mm is obtained by changing the field of $\pm 6\%$ but the wiggler tuning range is larger than ± 3.0 mm. The wiggler dispersion is self-matched while a small variation of R_{56} must be properly compensated by the DL quadrupoles.

It is worth remarking that the overall length of the DL is a very critical parameter. The tuning capability is limited to $\sim \pm 3.0$ mm so that an accurate estimate of the trajectory length inside elements such as dipoles and wiggler must be done. Simulations based on magnetic measurements and focussed on the definition of the ‘real’ trajectories are required. Moreover an installation strategy for placing the elements with the required precision must be studied.

2.4. Injection Region

The injection scheme, briefly described in the introduction, is based on a 1.5 GHz RF deflector exerting a kick of $\theta_k = 10 \text{ mrad}$ on the odd bunches and $-\theta_k$ on the even ones. Figure 1 indicates also the presence of two small dipoles with bending angles of $\theta_k/2 = 5 \text{ mrad}$ necessary for making the injection-extraction scheme symmetric. The matching with the DL is completed by the two couples of septa of $2i$ and $27i.14$. It is worth to remark that the odd bunches receive by the deflector a second kick of θ_k when leaving the DL. The net deflection at the output of the DL injection region is 2π for the odd bunches and 0 for the even ones, so that both bunches are recombined in the same trajectory.

2.5. Even Bunches Line Matching

Figure 7 shows the optical functions that the sequences of even bunches will see through the DL injection region. Dispersion is self-matched and beta functions and derivatives present the same values of the ones see by the odd bunches at the end of the DL. The Isochronicity condition is not completely fulfilled but the very small value of R_{56} at the line output (0.2 mm) does not require any particular care.




Figure 7. Even Bunches Line Optical Functions.

3. OTHER REQUIREMENTS

3.1. Beam Envelope and Stay Clear Requirements

Table 7 shows the stay clear apertures of the DL elements. In the case of the Injection Dipoles and of the RF Deflector, the only elements not yet designed, the stay clear apertures are suggested.

Table 7. Delay Loop Elements Stay Clear.

Element	Horizontal [mm] (total)	Vertical [mm] (total)
Dipoles	136	42
Quadrupoles	100 (fl)	100 (fl)
Sextupoles	100 (fl)	100 (fl)
Wiggler	100	40
2 _i Septa	20 (fl)	20 (fl)
27 _i .14 Septa	20 (fl)	20 (fl)
Injection Dipoles *	> 35	> 35
RF Deflector *	> 35	> 35

* Suggested stay clear apertures.




Figure 8. Delay Loop Beam Envelopes (rms values).

Figure 8 shows the rms beam envelopes in half the DL. Envelopes have been calculated according to:

$$\sigma_w = \sqrt{\beta_w \varepsilon_w + (D_w \Delta p/p)^2} \quad w = x, y \quad (5)$$

where ε_w are the transverse emittances. The values used were $\varepsilon_x = \varepsilon_y = 10^{-6} m$ and $\Delta p/p = 0.833 \%$ (value consistent with the total momentum acceptance of 5 %).

The criteria of defining the vacuum chamber stay clear greater than ten times the rms beam envelope will be used. From Figure 8, the total apertures that the vacuum chamber must have all along the DL are at least 80 mm and 35 mm in the horizontal and vertical planes respectively. This condition can be fulfilled in the entire loop except the injection region. In particular the vacuum chamber inside both septa is a cylinder with 20 mm inner diameter. In the 2j septum this implies a stay clear 8 times the beam envelope, which is still a reasonable value. In the 27j.14 septum horizontal plane the beam envelope increases from the beginning to the end in such a way that the stay clear passes from 7.7 to 5.4 times the beam envelope. In that point the beam size is dominated by the horizontal dispersion due to the septa magnets that cannot be reduced. The situation seems to indicate that the design of a new vacuum chamber for the 27j.14 septum (and maybe of a new septum as well) is required. The suggested total stay clear for the new element could be 30 mm for both the planes.

3.2. Considerations on Orbit Correction.

The DL orbit must be carefully corrected in order to minimize path length orbit modulation and the effects of non-linearities on the isochronicity. A very conservative solution should include one beam position monitors (BPM) per quadrupole, 34 in the DL case, while a reasonable compromise could have four BPM's per betatron period that means in the DL about 24 BPM's. Of course, the decision about the number of BPM's must take into account also the maximum impedance that the vacuum chamber can present to the beam.

As far as the number of corrector magnets is concerned, at least one horizontal corrector downstream each of the bending magnets, the wiggler and the 27.14j septa should be foreseen (12-13 correctors). The vertical plane is less demanding (no dipoles; small vertical tune) and one vertical corrector every two horizontal ones should ensure an effective orbit correction.

3.3. Final Remarks.

In order to properly define vacuum chamber shape and materials, the effects of beam impedance, that can affect in particular the bunch energy spread, must be investigated and an impedance budget for the DL must be defined. In particular, for high peak current bunches such as the DL ones, special attention must be paid in evaluating the effects of coherent synchrotron radiation.

The last consideration will concern diagnostics. A couple of current monitors upstream and downstream the DL could be helpful in the tuning phase while a synchrotron radiation port in a transfer line bending magnet downstream the DL could be a very effective tool for the isochronicity tuning.

REFERENCES

- [1] CLIC Study Team, *Proposal for Future CLIC Studies and a New CLIC Test Facility (CTF3)*, CERN/PS 99-047 (LP) and CLIC Note 402 (1999).
- [2] C. Biscari et al., *CTF3: Design of Driving Beam Combiner Ring*, 7th European Particle Accelerator Conference, Vienna 26-30 June 2000.
- [3] *LEP Preinjector Parameter List (EPA part)*, PS/LPI Note 88-02, March 1988 (rev.).
- [4] B. Bolli et al., *The Injection/extraction 2 deg Septum Magnets of the DAΦNE Accumulator*, DAΦNE Technical Note MM-7, June 7, 1995.
- [5] B. Bolli et al., *Measurements on Tesla Quadrupole Prototype for the DAΦNE Accumulator and Main Rings*, DAΦNE Technical Note MM-4, December 2, 1994.
- [6] B. Bolli et al., *The DAΦNE Accumulator Sextupoles*, DAΦNE Technical Note MM-6, May 10, 1995.
- [7] R. Chritin and D. Cornuet, SL-Note-2000-031 MS (2000).

APPENDIX. Delay Loop MAD8 Input File.

```

TITLE, s= "CTF3 DELAY LINE 4.1"

energ=200.

r56=0.00

! DRIFTS

ltsp1=.0
ltsp2=.0
ltsp3=.0

l2spt: drift, L=.601
l3spt: drift, L=.30
l4spt: drift, L=.15967830+ltsp1
l5spt: drift, L=.2569082333-ltsp1+ltsp2
l5bspt: drift, L=.08
l5aspt: drift, L=.0769082333-ltsp1+ltsp2
l6aspt: drift, L=.1
l6bspt: drift, L=.5714361667-ltsp2+ltsp3
l7spt: drift, L=.2614173-ltsp3

ltrim1=.0
ltrim2=.0
ltrim3=.0
ltrim4=.0

ls1cell: drift, L=.25911821+ltrim1
ls2cell: drift, L=.20367539-ltrim1+ltrim2
ls3acell: drift, L=.651629259-ltrim2+ltrim3
ls3bcell: drift, L=.1
ls4cell: drift, L=.4382139622-ltrim3+ltrim4
ls5cell: drift, L=.2985063488-ltrim4

ltr1=.0
ltr2=.0
ltr3=.0
ltr4=.0

l1cell: drift, L=.710321522+ltr1
l2cell: drift, L=.458436728-ltr1+ltr2
l2acell: drift, L=.1
l2bcell: drift, L=.258436728-ltr1+ltr2
l3cell: drift, L=.62663305-ltr2+ltr3
l3acell: drift, L=.1
l3bcell: drift, L=.42663305-ltr2+ltr3
l4cell: drift, L=.18149379-ltr3+ltr4
l5cell: drift, L=.28708954-ltr4

ltmt1=.0
ltmt2=.0
ltmt3=.0

lmt1 :drift, L=.2742733905+ltmt1
lmt2 :drift, L=.5206218105-ltmt1+ltmt2
lmt3 :drift, L=.1663871905-ltmt2+ltmt3
lmt4 :drift, L=.4815541305-ltmt3

```

! QUADRUPOLES

lq=.30

ksp1=-2.580128+r56*(-3.8182)
ksp2=+7.427991+r56*1.3549
ksp3=-4.614822+r56*7.0190
ksp4=+4.095144+r56*1.9261

kq1=+1.302465+r56*(-2.0670)
kq2=+7.031947+r56*0.0468
kq3=-7.006598+r56*2.4419
kq4=+6.680718+r56*(-0.6186)

kq5=+5.652653+r56*(-2.6396)
kq6=-5.140066+r56*0.4981
kq7=+7.809153+r56*3.1592

kmt1=0.5559753+r56*(-0.34322)
kmt2=-3.367295+r56*(-0.3521)
kmt3=4.587615+r56*(-2.2127)

q1spt: QUAD,l=lq,k1=ksp1
q2spt: QUAD,l=lq,k1=ksp2
q3spt: QUAD,l=lq,k1=ksp3
q4spt: QUAD,l=lq,k1=ksp4

cq1a: QUAD,l=lq/2,k1=kq1
cq2: QUAD,l=lq,k1=kq2
cq3: QUAD,l=lq,k1=kq3
cq4: QUAD,l=lq,k1=kq4

cq5: QUAD,l=lq,k1=kq5
cq6: QUAD,l=lq,k1=kq6
cq7: QUAD,l=lq,k1=kq7

qmt1a: QUAD,l=lq/2,k1=kmt1
qmt2: QUAD,l=lq,k1=kmt2
qmt3: QUAD,l=lq,k1=kmt3

! SEXTUPOLES

ksd0=-189.7212
ksd1=0
ksd2=-69.44286
ksd3=-26.81291
ksd4=-143.4109

lsex=.1

sd0: SEXTUPOLE,L=lsex,k2=ksd0
sd1: SEXTUPOLE,L=lsex,k2=ksd1
sd2: SEXTUPOLE,L=lsex,k2=ksd2
sd3: SEXTUPOLE,L=lsex,k2=ksd3
sd4: SEXTUPOLE,L=lsex,k2=ksd4


```

! DIPOLES

!   EPA magnetic measurement data

nbend  := 12
abend  := 6.28318531/nbend
hgap   := 0.0225

! measurements by G. Suberlucq, 1985 scaled to energy and number of dipoles
! fint  := 0.557
! lBend := 0.5629   ; Bin   := 0.65371*(energ/500)*(16/nbend)
! lGrad := 0.4666   ; Gin   := -0.4625*(energ/500)*(16/nbend)
! lSext := 0.0556   ; Sin   := 0.1*(energ/500)*(16/nbend)
!
! measurements by Chritin and Cornuet @ 400 MeV, SL-Note-2000-031 MS (2000)

fint   := 0.561
lBend  := 0.5648   ; Bin   := 0.52485*(energ/400)*(16/nbend)
lGrad  := 0.4681   ; Gin   := -0.37157*(energ/400)*(16/nbend)
lSext  := 0.6154   ; Sin   := 0.80*(energ/400)*(16/nbend)

b0      := bin / lbend
g0      := gin / lgrad
s0      := sin / lsext
mybrho  := bin / abend

myk1    := g0 / mybrho
myk2    := Sin / mybrho

esext   := s0*(lsext - lbend) / (2*b0) * lbend * abend / nbend
equad   := g0*(lbend - lgrad) / (2*b0)
e12     := equad + esext

b:   rbend, l=lbend,angle=-abend,k1=myk1,hgap=hgap,fint=fint,e1=-e12,&
      e2=-e12
bm1: rbend, l=lbend/2,angle=- abend/2 ,k1= myk1 ,hgap=hgap,fint=fint/2,&
      e1= -abend/4-e12, e2=abend/4.
bm2: rbend ,l=lbend/2,angle=- abend/2 ,k1= myk1 ,hgap=hgap,fint=fint/2,&
      e1= abend/4, e2=-abend/4-e12

syb : multipole, k2l = -2*myk1*e12 + myk2/2

!lh : drift, l = 0.742 - lbend/2
!bmod: line = (lh,syb,b,syb,lh)
!bmod1: line= (syb,bm1) !line = (lh,syb,bm1)
!bmod2: line= (bm2,syb) !line = (bm2,syb,lh)

bmod: line = (syb,b,syb)
bmod1: line= (syb,bm1)
bmod2: line= (bm2,syb)

! WIGGLER

angwig = 7.0*pi/180

lwig0 = 0.400996819
lwig  = 0.40*(angwig/sin(angwig))

```

```

bc1: sbend,l=lwig, angle=-angwig,e1=0,e2=-angwig
bc2: sbend,l=lwig, angle=angwig,e1=angwig,e2=0.
bc3: sbend,l=lwig, angle=angwig,e2=angwig,e1=0.
bc4: sbend,l=lwig, angle=-angwig,e1=-angwig,e2=0

wig1:line=(bc1,bc2)
wig2:line=(bc3,bc4)
wiggler:line=(wig1,wig2)

! SEPTA

rspt1=17.7518
rspt2=2.077814
sptlang=2*pi/180
spt2ang=27.14*pi/180
lspt1=rspt1*sptlang
lspt2=rspt2*spt2ang

spt1: sbend, l= lspt1, angle=-sptlang
spt2: sbend, l= lspt2, angle=-spt2ang

! KICKER

lkck=.5
kangle=0.01

kckp: sbend, l= lkck, angle=kangle
kckn: sbend, l= lkck, angle=-kangle

! SMALL INJECTION DIPOLE

ldipoletto=.3

dipoletto: sbend,l=ldipoletto, angle=-kangle/2

! VARIOUS

mq1: marker
mq2: marker
mh: marker
mb1: marker
mb2: marker

!##### BEAM LINES #####

! SEPTUM LINES

l1spt: drift, L=1.-lkck/2

sptin: line (kckn,l1spt,spt1,l2spt,spt2,l3spt,q1spt,l4spt,&
            q2spt,l5aspt,sd0,l5bspt,q3spt,l6aspt,sd1,l6bspt,q4spt,&
            l7spt,mb1,bmod1)

sptout: line (bmod2,mb2,l7spt,q4spt,l6bspt,sd1,l6aspt,q3spt,&
            l5bspt,sd0,l5aspt,q2spt,l4spt,q1spt,l3spt,spt2,l2spt,spt1&
            ,l1spt,kckn)

```

```

! INPUT AND OUTPUT LINES

lh1: drift, L=lspt1+.7123+lkck/2
li1: drift, L=.2
li2: line (l1spt,l2spt,lh1)

lin: line(li1,dipoletto,li2)

lout: line(-lin)

! QUARTERS

agspti: line(lin,sptin)
qu2spti: line(bmod2,ls1cell,cq5,ls2cell,cq6,ls3acell,sd2,&
             ls3bcell,cq7,ls4cell,bmod1,bmod2,ls5cell,cq1a)

quarterin: line(agspti,qu2spti)

quli: line(cq1a,l1cell,bmod1,bmod2,l2cell,cq2,l3acell,sd3,&
           l3bcell,cq3,l4cell,cq4,l5cell,bmod1)
qu2i: line(bmod2,l5cell,cq4,l4cell,cq3,l3cell,cq2,l2acell,sd4&
           ,l2bcell,bmod1,bmod2,lmt1,qmt1a,qmt1a,lmt2,qmt2,lmt3,&
           qmt3,lmt4,wig1)

quarter1: line(quli,qu2i)

qu2o: line(wig2,lmt4,qmt3,lmt3,qmt2,lmt2,qmt1a,qmt1a,lmt1&
           ,bmod1,bmod2,l2bcell,sd4,l2acell,cq2,l3cell,cq3,&
           l4cell,cq4,l5cell,bmod1)
qu1o: line(bmod2,l5cell,cq4,l4cell,cq3,l3bcell,sd3,l3acell,&
           cq2,l2cell,bmod1,bmod2,l1cell,cq1a)

quarter2: line(qu2o,qu1o)

qu2spto: line(cq1a,ls5cell,bmod1,bmod2,ls4cell,cq7,ls3bcell,&
             sd2,ls3acell,cq6,ls2cell,cq5,ls1cell,bmod1)
agspto: line(sptout,lout)

quarterout: line(qu2spto,agspto)

! HALFS

half1: line(quarterin,mq1,quarter1)

half2: line(quarter2,mq2,quarterout)

! DELAY

delay: line(half1,mh,half2)

! EVENLINE

evenline: line(lin,kckp,lout)

```

```

! #####

b0x=4.498751
b0y=6.0

bxa= b0x*(1+3.683*3.683/b0x/b0x)
bya= b0y*(1+3.683*3.683/b0y/b0y)
alfxa=3.683/b0x
alfya=3.683/b0y
dxa=0.
dpxa=0.

useline: line(delay)

BEAM, mass=0.51098e-3,energy=energ/1000,Ex=1e-6,Ey=1e-6,&
      sige=.0083333333

USE,useline

match, betx=bxa,bety=bya,alfx=alfxa,alfy=alfya,dx=dxa,dpx=dpxa,deltap=0.

!vary, ksp1,step=.001,lower=-9.,upper=9.
!vary, ksp2,step=.001,lower=-9.,upper=9.
!vary, ksp3,step=.001,lower=-9.,upper=9.
!vary, ksp4,step=.001,lower=-9.,upper=9.
!vary, kq1,step=.001,lower=0.,upper=9.
!vary, kq2,step=.001,lower=0.,upper=9.
!vary, kq3,step=.001,lower=-9.,upper=0.
!vary, kq4,step=.001,lower=0.,upper=9.
!vary, kq5,step=.001,lower=0.,upper=9.
!vary, kq6,step=.001,lower=-9.,upper=0.
!vary, kq7,step=.001,lower=0.,upper=9.
!vary, kmt1,step=.001,lower=-9.,upper=9.
!vary, kmt2,step=.001,lower=-9.,upper=9.
!vary, kmt3,step=.001,lower=-9.,upper=9.
!vary, b0x,step=.01,lower=2.5,upper=10
!vary, b0y,step=.01,lower=2.,upper=6
!vary, bxa,step=.01, lower=.3,upper=15
!vary, bya,step=.01, lower=2,upper=15
!vary, dxa,step=.01, lower=-2,upper=-0.05
!vary, alfxa,step=.01, lower=0,upper=+10
!vary, alfy,step=.01, lower=0,upper=+10
!vary, ltmt1,step=.01, lower=-0.11,upper=.2
!vary, ltmt2,step=.01, lower=-0.2,upper=.08
!vary, ltmt3,step=.01, lower=-0.15,upper=.11
!vary, ltr1,step=.01, lower=-0.15,upper=.0
!vary, ltr2,step=.01, lower=-0.15,upper=.15
!vary, ltr3,step=.01, lower=-0.15,upper=.15
!vary, ltr4,step=.01, lower=-0.15,upper=.15
!vary, ltrim1,step=.01, lower=-0.6,upper=.0
!vary, ltrim2,step=.01, lower=-0.2,upper=.2
!vary, ltrim3,step=.01, lower=-0.15,upper=.05
!vary, ltrim4,step=.01, lower=-0.15,upper=.05
!vary, ltsp1,step=.01, lower=-0.28,upper=.28
!vary, ltsp2,step=.01, lower=-0.28,upper=.28
!vary, ltsp3,step=.01, lower=-0.28,upper=.28
!vary, ksd0,step=.1, lower=-195,upper=195
!vary, ksd1,step=.1, lower=-195,upper=195
!vary, ksd2,step=.1, lower=-195,upper=195
!vary, ksd3,step=.1, lower=-195,upper=195
!vary, ksd4,step=.1, lower=-195,upper=195

```

```

weight ,dx=.1,dpx=.1,alfx=.1,alfy=.1,betx=.1,bety=.1
constraint,mq1,dx=0
constraint,mq1,dpx=0
constraint,mq2,dx=0
constraint,mq2,dpx=0
constraint,#E,dx=0
constraint,#E,dpx=0.
constraint,#E,alfx=-alfxa
constraint,#E,alfy=-alfya
constraint,#E,betx=bxax
constraint,#E,bety=byax
rmatrix,range=useline,rm(5,6)=r56
tmatrix, range=useline,tm(5,6,6)=0.
tmatrix, range=useline,tm(5,1,6)=0.
tmatrix, range=useline,tm(5,2,6)=0.
tmatrix, range=useline,tm(5,3,6)=0.
tmatrix, range=useline,tm(5,4,6)=0.
simplex,calls=100000
migrad,calls=100000
endmatch

print, full

twiss,betx=bxax,bety=byax,alfx=alfxa,alfy=alfya,&
  dx=dxax,dpx=dpxax,deltap=0.,tape,save

plot,haxis=s,vaxis=betx,bety,dx,spline

!survey, x0=0.,y0=0.,z0=-3.682824,theta0=0.,tape

TRACK,ONEPASS

START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=0.
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=0.005
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=0.01
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=0.015
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=0.02
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=0.025
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=-0.005
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=-0.01
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=-0.015
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=-0.02
START, x=0.,y=0.,px=0.,py=0.,T=0.,deltap=-0.025

RUN,turns=1,table=trck1

!PLOT,symbol=3,table=trck1,Haxis=deltap,vaxis=T,&
!   vmin=-0.025,vmax=.005,&
!   particle=1,2,3,4,5,6,7,8,9,10,11,multiple

ENDTRACK

!sigx:string,"sqrt(beam[ex]*betx+dx*dx*beam[sigx]*beam[sigx])"
!sigy:string,"sqrt(beam[ey]*bety)"
!emx:string,"beam[ex]"
!emy:string,"beam[ey]"
!enspr:string,"beam[sigx]"!!table, name=twiss, columns=s,sigx,sigy
!archive, filename=prova.dat
!plot,table=twiss,haxis=s,vaxis=sigx,sigy,spline

STOP

```